

Prolog

- *Scopo:*
 1. Acquisire conoscenza dell'ambiente TuProlog/SWI Prolog
 2. Risolvere alcuni problemi mediante il linguaggio Prolog
 - Prendere dimestichezza con un approccio DICHIARATIVO
 - Ricorsione e ricorsione tail
 - Liste
 - Cut

Istruzioni su SWI

Avvio di SWI Prolog

L'interprete può essere avviato direttamente dal menù Avvio di Windows. All'avvio il sistema si identifica e presenta il proprio prompt, con un output del tipo:

```
| ?-
```

Directory di lavoro corrente

SWI parte facendo direttamente riferimento ad una “directory di lavoro corrente”. Per sapere qual'è, si possono usare i comandi:

- `pwd`.
- `working_directory(Old, Old)` .

Nota: “Old” scritto con la maiuscola è una variabile!

Istruzioni su SWI

Per cambiare directory di lavoro

- `working_directory(_, New).`

Dove al posto di “New” dovete mettere la directory desiderata. Ad es.:

```
working_directory(_, 'd:/fchesani').
```

Caricamento di programmi

I programmi Prolog possono essere caricati o tramite il menù “File→consult...”, o mediante il predicato “consult” direttamente al prompt:

```
?- consult(flatten).
```

```
% flatten compiled 0.00 sec, 1,492 bytes
```

```
true.
```

```
?-
```

(si noti che non è necessario specificare l'estensione del file: SWI cerca automaticamente file con estensione “.pl”)

Istruzioni su SWI

Tracing

- Il comando “**trace**” consente di seguire passo passo la risoluzione delle clausole; tale modalità può poi essere disabilitata con il comando “**notrace**”.

Per eseguire il trace (debug) di una query, è necessario invocare il predicato `trace`, seguito subito dopo dal goal che si vuole verificare. Ad esempio:

```
?- trace, p(X).
```

Permette di eseguire passo passo la risoluzione della query `p(X)`.

SWI – versione online

E' disponibile anche una versione online dell'interprete SWI:

<https://swish.swi-prolog.org/>

TuProlog (2P)

- Sviluppato da colleghi del DISI (capo progetto: Prof. Enrico Denti)
- Sito principale:
<http://apice.unibo.it/xwiki/bin/view/Tuprolog/>
- Link per il download (per questa esercitazione):
<https://bitbucket.org/tuprologteam/tuprolog/downloads>

Caratteristiche:

- Pure Java
- Dispone di una interfaccia grafica...
 - oggi useremo quest'ultima... “eseguibile”: “2p.jar”
- Disponibile anche per Android, .NET, e come plugin per ambiente Eclipse
- Home page: <http://apice.unibo.it/xwiki/bin/view/Tuprolog/>

TuProlog (2P)

The image shows the TuProlog IDE interface. At the top, there is a menu bar with icons for file operations (New, Open, Save, Print, Undo, Redo, Cut, Copy, Paste, Find) and a status bar showing "Line: 1". Below the menu bar is a toolbar with icons for running, stepping through, and other execution controls. The main workspace is a large text area labeled "untitled" with a yellow background, currently empty. Below the workspace is a query input field containing "?-" and a "Query..." label pointing to it. Underneath the query field are tabs for "solution", "bindings", "all bindings", "output", and "exceptions". The "solution" tab is active, showing "no.". At the bottom, there is a control panel with buttons for "Next", "Accept", "Stop", "Clear", and "Export GSV". A green box highlights the "Next" button, with a label "Altre soluzioni..." pointing to it. The status bar at the very bottom displays "No. Ready."

Editor del programma prolog

Query...

Soluzioni, binding delle variabili, etc.

Altre soluzioni...

TuProlog (2P)

Alcune osservazioni:

- Alcuni predicati “classici” sono già definiti, e non necessitano di essere “caricati” tramite apposita libreria
 - E.g., `member(EL, LIST)`
- Non c'è il concetto di working directory (come accade invece in altri prolog...)
- E' disponibile una finestra grafica per il debug (miglior supporto disponibile nel plug-in per Eclipse)
- Necessario salvare i propri programmi in file appositi...

Es. 1

Si definisca un predicato in PROLOG chiamato **maxlist** che applicato ad una lista di liste di interi **ListListInt** dia come risultato la lista degli elementi massimi di ogni lista componente di **ListListInt**. Si definisca prima la versione ricorsiva e poi quella ricorsiva-tail.

Esempio:

?-

```
?- maxlist([[3,10,2], [6,9], [1,2]], X).
```

```
yes, X = [10,9,2]
```

Es. 1 - Soluzione

```
maxlist([], []).
```

```
maxlist([X|Y], [N|T]) :- max(X, N), maxlist(Y, T).
```

Versione ricorsiva:

```
max([X], X) :- !.
```

```
max([X|T], X) :- max(T, N), X >= N, !.
```

```
max([X|T], N) :- max(T, N).
```

Es. 1 - Soluzione

```
maxlist([], []).
```

```
maxlist([X|Y], [N|T]) :- max(X, N), maxlist(Y, T).
```

Versione ricorsiva:

```
max([X], X) :- !.
```

```
max([X|T], X) :- max(T, N), X >= N, !.
```

```
max([X|T], N) :- max(T, N).
```

Versione iterativa:

```
max([X|T], M) :- max(T, X, M).
```

```
max([], M, M) :- !.
```

```
max([H|T], MT, M) :- H > MT, !, max(T, H, M).
```

```
max([H|T], MT, M) :- max(T, MT, M).
```

Es.1 – Uso del !

- Il ! deve essere usato con scrupolosa attenzione
- Consideriamo il programma Prolog

```
max ([X] , X) :- ! .
max ([X|T] , X) :- max (T, N) , X>=N, ! .
max ([X|T] , N) :- max (T, N) .
```
- Il programma **funziona correttamente** quando invocato con secondo parametro (ovvero il massimo) VARIABILE
 - Es.: `max ([1, 2, 4, 3] , M)`
- Il programma **NON** funziona correttamente quando invocato per testare se un certo valore è il massimo
- In particolare, poiché la terza clausola non contiene la condizione $X < N$, il programma restituisce *true*
 - Sia mettendo il valore massimo, es. `max ([1, 2, 4, 3] , 4)`
 - Sia mettendo l'ultimo valore contenuto nella lista, es. `max ([1, 2, 4, 3] , 3)`
- Si provino a disegnare gli alberi SLD per vedere cosa accade!

Es. 2

Data una lista **L1** e un numero intero **N**, scrivere un predicato Prolog **domanda1 (L1, N, L2)** che restituisca in **L2** la lista degli elementi di **L1** che sono **liste** contenenti **solo due valori** interi positivi fra 1 e 9 la cui somma valga **N**.

Esempio:

```
:- domanda1 (
    [ [3,1], 5, [2,1,1], [3], [1,1,1], a, [2,2] ],
    4,
    L2) .
yes, L2 = [[3,1], [2,2]]
```

Es. 2 - Soluzione

```
domanda1 ([ ], _, [ ] ).
domanda1 ( [[A,B] | R ], N, [[A,B] | S] ) :-
    A >= 1, A < 9, B >= 1, B < 9,
    N is A + B, !,
    domanda1 ( R, N, S ).
domanda1 ( [_ | R ], N, S ) :- domanda1 ( R, N, S ).
```

Es. 3

Si scriva un predicato Prolog che data una lista ed un elemento **E1** appartenente alla lista, restituisca in uscita l'elemento **successivo** ad E1 nella lista.

Esempio:

```
?- consec(3, [1, 7, 3, 9, 11], X) .
```

```
yes    X=9
```

Nel caso in cui E1 sia l'ultimo elemento il predicato dovrà fallire

Es. 3 - Soluzione

```
consec(E1, [E1| [X|_]],X) :- !.
```

```
consec(E1, [_|Tail],X) :- consec(E1,Tail,X).
```

Es. 4 – tratto dal Compito del 29 marzo 2006

Un giorno il mago rosso venne sfidato dal mago verde ad un duello di veleni. Ciascuno dei due doveva portare il veleno più potente che era riuscito a produrre; ciascuno avrebbe bevuto prima il veleno dell'altro e poi il proprio.

Vale infatti la regola che se si beve un veleno e poi uno più potente, allora non si muore, ma il secondo fa da antidoto per il primo.

Il mago rosso accettò la sfida, sapendo che il mago verde aveva un veleno molto più potente del suo: quello del mago verde ha potenza 8, mentre il mago rosso ha solo dell'acqua (che non è un veleno e ha potenza 0) ed un veleno a potenza 1. Il mago rosso, però ha pensato di bere qualcosa *prima* della sfida ...

Es. 4 – tratto dal Compito del 29 marzo 2006

Si scriva un predicato Prolog `vivo/1` che prende in ingresso una lista di numeri ed ha successo se bevendo quella sequenza di veleni si **sopravvive**.

Es:

```
?- vivo([1,8,0,1,4]).
```

yes

```
?- vivo([1,2,4,1]).
```

no

Es. 4 – Soluzione

```
vivo([]).  
vivo([0|L]):- !, vivo(L).  
vivo([A,B|L]):- A < B, vivo(L).
```

Es. vari

5. Scrivere un predicato `flatten` che “appiattisce” una lista di liste. Ad esempio:

```
:- flatten([ 1, a, [2, 3], [], h, f(3), [c, [d, [e]]] ], L) .  
yes, L=[1, a, 2, 3, h, f(3), c, d, e]
```

Suggerimento: cercare di riformulare il problema in maniera dichiarativa. Nella formulazione, assumere che “l’appiattimento di un elemento è una lista contenente quell’elemento”

6. Data una matrice $N \times N$ rappresentata come lista di liste, si calcoli la somma degli elementi della diagonale principale.
- Provare a ragionare in termini di “sottomatrici”
7. Si definisca un predicato Prolog che calcola i massimi locali (esclusi gli estremi) di una lista, ad esempio:

```
:- max_loc([5, 4, 7, 2, 3, 6, 1, 2], X)  
yes, X=[7, 6]
```

Es. 8 – posizione in una lista

Si scriva un predicato Prolog :

```
listPos(L, El, Pos)
```

che ricevuta in ingresso una lista L, restituisce la posizione Pos dell'elemento El specificato. Le posizioni si contano a partire da zero.

Ad esempio:

```
?- listPos([1,2,3,5,6,3], 3, X) .
```

```
yes .
```

```
X / 2
```

```
Solution: listPos([1,2,3,5,6,3],3,2)
```

Es. 8 – posizione in una lista

```
listPos([X|_], X, 0).  
listPos([_|T], X, Pos) :-  
    listPos(T, X, Pos1),  
    Pos is Pos1+1.
```

Questa funzione è ricorsiva tail?

Es. 8 – posizione in una lista

Ricorsiva tail:

```
%%% listPos/3
listPos(L, El, Pos) :- listPos(L, El, Pos, 0).

%%% listPos/4
%%% listPos(List, El, Pos, Part)
listPos([El|_], El, Pos, Pos).
listPos([_|T], El, Pos, Part) :-
    Part1 is Part + 1,
    listPos(T, El, Pos, Part1).
```

Compito del 16 Giugno 2016

Si scriva un predicato Prolog :

`list_int(M,N,L)`

che genera la lista L di tutti i numeri compresi fra M e N. Se $M > N$ genera la lista vuota.

Ad esempio:

```
?- list_int(3,5,L) .
```

```
L= [3,4,5]
```

Compito del 16 Giugno 2016 - Soluzione

```
list_int(M,N, []) :-  
    M > N.  
list_int(M,N, [M|Tail]) :-  
    M =< N,  
    M1 is M+1,  
    list_int(M1,N,Tail).
```

Questa funzione è ricorsiva tail?

Compito del 8 Settembre 2016

Si scriva un predicato Prolog

```
remove_list( L1, L2, L3)
```

che data una lista L1 e una lista L2 restituisca una lista L3 contenente gli elementi di L1 non contenuti in L2.

Esempi:

```
?- remove_list([a,b,a,b,b,c], [a,c], L) .  
L=[b,b,b] .
```

```
?- remove_list([a,b], [], L) .  
L=[a,b] .
```

Si definiscano tutti i predicati utilizzati.

Compito del 8 Settembre 2016 - Soluzione

```
remove_list( [], _, [] ) :- !.  
remove_list( [H|D], L2, R) :-  
    member(H, L2),  
    !,  
    remove_list(D, L2, R).  
remove_list( [H|D], L2, [H|R]) :-  
    remove_list(D, L2, R).  
  
member( X, [X|_] ) :- !.  
member( X, [_|D] ) :- member(X,D).
```

Questa funzione è ricorsiva tail?